Knowledge Representation and Reasoning

Exercises on ASP

1 Positive Programs

Determine the models of the following positive logic programs and decide which models are stable.

a) $$P = \left\{ \begin{aligned} rain &\leftarrow \\ wet &\leftarrow rain \\ wet &\leftarrow sprinkler \end{aligned} \right\}$$

b) $$P = \left\{ \begin{aligned} coffee &\leftarrow \\ lemon &\leftarrow tea \\ sugar &\leftarrow coffee \\ milk &\leftarrow coffee, sugar \\ tea &\leftarrow lemon \\ tea &\leftarrow diet \end{aligned} \right\}$$

c) $$P = \left\{ \begin{aligned} shirt &\leftarrow \\ sneakers &\leftarrow \\ pants &\leftarrow sneakers \\ skirt &\leftarrow shirt, sandals \\ sandals &\leftarrow dress \end{aligned} \right\}$$

d) $$P = \left\{ \begin{aligned} red &\leftarrow \\ meat &\leftarrow cabbage \\ meat &\leftarrow red \\ fish &\leftarrow asparagus \\ asparagus &\leftarrow fish, white \\ white &\leftarrow fish \end{aligned} \right\}$$

Answer:
a) Models

- $\{rain, wet\}$
- $\{rain, wet, sprinkler\}$

Stable Model: $\{rain, wet\}$
b) Models

- $\{coffee, sugar, milk\}$
- $\{coffee, sugar, milk, tea, lemon\}$
- $\{coffee, sugar, milk, tea, lemon, diet\}$

Stable Model: $\{coffee, sugar, milk\}$
c) and d) left as exercise

2 Positive Programs with Variables

Determine the Herbrand models of the following positive logic programs with variables and decide which Herbrand models are stable.

a) $$\mathrm{P} = \begin{Bmatrix} fish(blinky) \leftarrow \\ bird(tweety) \leftarrow \\ flies(X) \leftarrow bird(X) \end{Bmatrix}$$

b) $$\mathrm{P} = \begin{Bmatrix} next(0,1) \leftarrow \\ next(1,0) \leftarrow \\ even(0) \leftarrow \\ even(Y) \leftarrow next(X,Y), odd(X) \\ odd(Y) \leftarrow next(X,Y), even(X) \end{Bmatrix}$$

c) $$\mathrm{P} = \begin{Bmatrix} friend(alice, bob) \leftarrow \\ friend(bob, alice) \leftarrow \\ friend(eve, alice) \leftarrow \\ invite(alice) \leftarrow \\ invite(Y) \leftarrow invite(X), friend(X,Y) \end{Bmatrix}$$

d) $$\mathrm{P} = \begin{Bmatrix} next(0,1) \leftarrow \\ next(1,2) \leftarrow \\ before(X) \leftarrow next(X,Y) \\ between(Y) \leftarrow next(X,Y), before(Y) \end{Bmatrix}$$

Answer: a) Herbrand Models:

- $\{fish(blinky), bird(tweety), flies(tweety)\}$
- $\{fish(blinky), bird(tweety), flies(tweety), flies(blinky\}$
- $\{fish(blinky), bird(tweety), flies(tweety), fish(tweety)\}$
- $\{fish(blinky), bird(tweety), flies(tweety), fish(tweety), flies(blinky)\}$
- $\{fish(blinky), bird(tweety), flies(tweety), bird(blinky), flies(blinky)\}$
- $\{fish(blinky), bird(tweety), flies(tweety), bird(blinky), flies(blinky), fish(tweety)\}$

Stable Herbrand Model: $\{fish(blinky), bird(tweety), flies(tweety)\}$
b) Herbrand Models:

- $\{next(0,1), next(1,0), even(0), odd(1)\}$
- $\{next(0,1), next(1,0), even(0), odd(1), even(1), odd(0)\}$
- $\{next(0,1), next(1,0), even(0), odd(1), even(1), odd(0), next(0,0)\}$
- $\{next(0,1), next(1,0), even(0), odd(1), even(1), odd(0), next(1,1)\}$
- $\{next(0,1), next(1,0), even(0), odd(1), even(1), odd(0), next(0,0), next(1,1)\}$

Stable Herbrand Model: $\{next(0,1), next(1,0), even(0), odd(1)\}$
c) and d) left as an exercise

3 Normal Programs

Determine the models of the following normal logic programs and decide which models are stable.

a) $$P = \left\{ \begin{array}{r} sprinkler \leftarrow \sim rain \\ rain \leftarrow \sim sprinkler \\ wet \leftarrow rain \\ wet \leftarrow sprinkler \end{array} \right\}$$

b) $$P = \left\{ \begin{array}{r} diet \leftarrow \sim sugar \\ coffee \leftarrow \sim tea \\ lemon \leftarrow tea \\ sugar \leftarrow coffee \\ milk \leftarrow coffee, sugar \\ tea \leftarrow lemon \\ tea \leftarrow diet \end{array} \right\}$$

c) $$P = \left\{ \begin{array}{r} dress \leftarrow \sim shirt \\ shirt \leftarrow \sim dress \\ sandals \leftarrow \sim sneakers \\ sneakers \leftarrow \sim sandals \\ pants \leftarrow sneakers \\ skirt \leftarrow shirt, sandals \\ sandals \leftarrow dress \end{array} \right\}$$

d) $$P = \left\{ \begin{array}{r} asparagus \leftarrow \sim cabbage \\ cabbage \leftarrow \sim asparagus \\ red \leftarrow \sim white \\ meat \leftarrow cabbage \\ meat \leftarrow red \\ fish \leftarrow asparagus \\ asparagus \leftarrow fish, white \\ white \leftarrow fish \end{array} \right\}$$

Answer: a) Models:
$\{sprinkler, wet\}$, $\{rain, wet\}$, and $\{sprinkler, rain, wet\}$
Stable Models: $\{sprinkler, wet\}$ and $\{rain, wet\}$
b) Models:

- $\{coffee, sugar, milk\}$
- $\{coffee, sugar, milk, lemon, tea\}$
- $\{coffee, sugar, milk, diet, lemon, tea\}$
- $\{diet, tea, lemon\}$
- $\{diet, tea, lemon, milk\}$
- $\{diet, tea, lemon, sugar\}$
- $\{diet, tea, lemon, milk, sugar\}$
- $\{tea, sugar, lemon\}$
- $\{tea, sugar, lemon, milk\}$

Stable Models: $\{coffee, sugar, milk\}$ and $\{diet, tea, lemon\}$
c) and d) left as an exercise

4 Answer Set Solving

The following exercises can be modelled/implemented and tested using clingo. The answer set solver can be downloaded and installed from here:

`https://potassco.org/`

This web page also contains information on how to set up the system and the guide on its usage. In case you did not bring your computer (clingo is not installed in the lab), you can use the browser version:

`https://potassco.org/clingo/run/`

1) Install clingo on your computer (alternatively, start the browser version) and make yourself familiar with the system.

5 The Boat Trip Again ...

Consider the following situation:

> *"John wants to make a boat trip with two friends: Peter and Mary. John's mother agrees with this as long as there is a guarantee that it is safe to travel by boat with at least one of John's companions. Moreover, John's mother finds it normal that to travel by boat is safe with the company of someone who has had a course in navigation.*
>
> *Meanwhile, John's father takes cognizance of the matter and, in that regard, tells John's mother he is sure at least one of those of John's friends has had a course in navigation, though he cannot recall who".*

1) Encode this knowledge in an answer set program such that a) we obtain two answer sets and agrees holds in both of them; b) we still obtain two answer sets if we learn that it is not safe to travel with Peter, but agrees does no longer hold in both of them.

Answer:

```
course(peter) :- not course(mary).
course(mary) :- not course(peter).
safeWith(X):-course(X), not unsafe(X).
agrees:-companion(X),safeWith(X).
companion(mary).
companion(peter).
#show agrees/0.
```

6 Exact Hitting Set

Given a collection of sets, the Exact Hitting Set problem is to select exactly one element from each set. For example, the sets $\{a, b, c\}$, $\{a, c, d\}$, and $\{b, c\}$ have two exact hitting sets: $\{b, d\}$ and $\{c\}$. We represent such a problem instance by facts as follows:

```
set(1). element(1,a). element(1,b). element(1,c). % {a,b,c}
set(2). element(2,a). element(2,c). element(2,d). % {a,c,d}
set(3). element(3,b). element(3,c).               % {b,c}

% Solutions: {select(b),select(d)}, {select(c)}
```

Specify a uniform problem encoding such that atoms over the predicate `select/1` within stable models correspond to exact hitting sets for arbitrary instances.

Answer: Several solutions and ways how to achieve a solution are possible. Three different ones are pointed out in the following.

- Using a normal program (with constraints):

```
% generate all possible combinations
select(X) :- element(_,X), not omitted(X).
omitted(X) :- element(_,X), not select(X).
% ensure that each set has at least one element
test(S):-set(S),element(S,X),select(X).
:- not test(S), set(S).
% ensure that there is at most one element per set
:- set(S), element(S,X), element(S,Y), select(X), select(Y), X!=Y.
% show only select predicates
#show select/1.
```

- Using an aggregate for generation together with constraints

```
% generate all possible combinations
{select(X):element(S,X)}:- set(S).
% ensure that each set has at least one element
test(S):-set(S),element(S,X),select(X).
:- not test(S), set(S).
% ensure that there is at most one element per set
:- set(S), element(S,X), element(S,Y), select(X), select(Y), X!=Y.
% show only select predicates
#show select/1.
```

- Using just an aggregate

```
% generate only valid combinations
1{select(X):element(S,X)}1:- set(S).
% show only select predicates
#show select/1.
```

7 Independent Set

Given an undirected graph, the Independent Set problem is to select a subset of vertices such that no pair of selected vertices is connected by an edge, where the number of selected vertices must be equal or greater than a given threshold. For example, the graph $(\{1,2,3,4,5,6\},\{\{1,2\},\{1,3\},\{2,4\},\{3,5\},\{4,5\},\{4,6\}\})$ has three independent sets with 3 vertices (at least): $\{1,5,6\}$,$\{2,5,6\}$, and $\{2,3,6\}$. We represent such a problem instance by facts as follows:

```
vertex(1). vertex(2). vertex(3). vertex(4). vertex(5). vertex(6).
edge(1,2). edge(1,3). edge(2,4). edge(3,5). edge(4,5). edge(4,6).
threshold(3).
% Solutions:  {select(1),select(5),select(6)},
%             {select(2),select(5),select(6)},
%             {select(2),select(3),select(6)}
```

Specify a uniform problem encoding such that atoms over the predicate `select/1` within stable models correspond to independent sets for arbitrary instances.

Answer:
```
Y{select(X):vertex(X)}Y:-threshold(Y).
:- select(X),select(Y),edge(X,Y),X!=Y.
#show select/1.
```

8 Vertex Cover

Given an undirected graph, the Vertex Cover problem is to select a subset of vertices such that each edge includes some selected vertex, where the number of selected vertices must not exceed a given threshold. For example, the graph $(\{1, 2, 3, 4, 5, 6\}, \{\{1, 2\}, \{1, 3\}, \{2, 4\}, \{3, 5\}, \{4, 5\}, \{4, 6\}\})$ has three vertex covers with 3 vertices (at most): $\{1, 3, 4\}$, $\{1, 4, 5\}$, and $\{2, 3, 4\}$. We represent such a problem instance by facts as follows:

```
vertex(1). vertex(2). vertex(3). vertex(4). vertex(5). vertex(6).
edge(1,2). edge(1,3). edge(2,4). edge(3,5). edge(4,5). edge(4,6).
threshold(3).
% Solutions:  {select(1),select(3),select(4)},
%             {select(1),select(4),select(5)},
%             {select(2),select(3),select(4)}
```

Specify a uniform problem encoding such that atoms over the predicate `select/1` within stable models correspond to vertex covers for arbitrary instances.

Answer:
```
{select(X):vertex(X)}Y:-threshold(Y).
:-not select(X), not select(Y), edge(X,Y).
#show select/1.
```

9 Dominating Set

Given an undirected graph, the Dominating Set problem is to select a subset of vertices such that each remaining vertex is adjacent to some selected vertex, where the number of selected vertices must not exceed a given threshold. For example, the graph $(\{1, 2, 3, 4, 5, 6\}, \{\{1, 2\}, \{1, 3\}, \{2, 4\}, \{3, 5\}, \{4, 5\}, \{4, 6\}\})$ has two dominating sets with 2 vertices (at most): $\{1, 4\}$ and $\{3, 4\}$. We represent such a problem instance by facts as follows:

```
vertex(1). vertex(2). vertex(3). vertex(4). vertex(5). vertex(6).
edge(1,2). edge(1,3). edge(2,4). edge(3,5). edge(4,5). edge(4,6).
threshold(2).
% Solutions: {select(1),select(4)}, {select(3),select(4)}
```

Specify a uniform problem encoding such that atoms over the predicate `select/1` within stable models correspond to dominating sets for arbitrary instances.

Answer:
```
{select(X):vertex(X)}Y:-threshold(Y).
test(X):-vertex(X), edge(X,Y), select(Y). test(X):-vertex(X), edge(Y,X), select(Y).
:- vertex(X), not select(X), not test(X).
#show select/1.
```

10 Stable Marriages

Given n men and n women, where each person has ranked all members of the opposite sex with a unique number between 1 and n in the order of preference, marry all men and women such that there are no two people of opposite sex who would both rather have each other than their current partners. For example, if Manuel is married to Paula, and Peter to Maria, and if Manuel prefers Maria to Paula and Maria prefers Manuel to Peter, then both their marriages are unstable. If there are no such people, all the marriages are *stable*.

The set of n men (resp. women) is given in the form of a set of facts $man(Name)$ (resp. $woman(Name)$).

Each man has his preferences about whom of the n women he would like to marry the most represented by $pref(NameMan, NameWoman, Value)$ where a larger number expresses a lower preference of the man for that woman. Similarly, for each woman preferences are represented by facts of the form $pref(NameWoman, NameMan, Value)$.

Present an answer set program such that each answer set corresponds exactly to one solution where every man and woman is stably married.

Answer:
```
1{married(X,Y):man(X)}1:-woman(Y).
:-married(X,Y), married(X,Z), Y!=Z.
:-married(X1,Y1), married(X2,Y2), pref(X1,Y1,V1),pref(X1,Y2,V2),
pref(Y2,X1,V3),pref(Y2,X2,V4), V2<V1,V3<V4.
#show married/2.
```

11 Stable Gay Marriages

Repeat the previous exercise, this time for gay couples. Given $2n$ people (male or female) where each has ranked all others with a unique number between 1 and 2n-1 in order of preference, present an answer set program such that each answer set corresponds exactly to one solution where every person is stably married.

Answer: Left as an exercise.

12 Stable Polygamous Marriages

Repeat the Exercise 10, this time for polygamous marriages. Given n men and m women, where each person has ranked all members of the opposite sex with a unique number between 1 and max(n,m) in the order of preference, and each man has expressed the maximum number of wives he accepts, present an answer set program such that each answer set corresponds exactly to one solution where there are no unstable marriages.

Answer: Left as an exercise.

13 Zebra Puzzle

Consider the following problem description.

1. There are five houses.
2. The Englishman lives in the red house.
3. The Spaniard owns the dog.
4. Coffee is drunk in the green house.
5. The Ukrainian drinks tea.
6. The green house is immediately to the right of the ivory house.
7. The Old Gold smoker owns snails.
8. Kools are smoked in the yellow house.
9. Milk is drunk in the middle house.
10. The Norwegian lives in the first house.
11. The man who smokes Chesterfields lives in the house next to the man with the fox.
12. Kools are smoked in a house next to the house where the horse is kept.
13. The Lucky Strike smoker drinks orange juice.

14. The Japanese smokes Parliaments.

15. The Norwegian lives next to the blue house.

Now, who drinks water? Who owns the zebra? In the interest of clarity, it must be added that each of the five houses is painted a different color, and their inhabitants are of different national extractions, own different pets, drink different beverages and smoke different brands of American cigarets [sic]. One other thing: in statement 6, right means your right.

Answer: Left as an exercise.